4.2 SIMPLE MODELING OF THE LEFT HEART AND SYSTEMIC ARTERIES

The modeling method discussed in Section 4.1 is usually satisfactory for veins and arteries, but the valves in the heart provide strong nonlinearities (or are at best only "piecewise" linear), while the muscles in the ventricular wall that provide the pumping action make this part of

the heart linear but time-varying. In this section we begin by developing a simple model of the left heart, consisting of mitral valve, ventricle, and aortic valve, with only a rudimentary systemic arterial load. Running this model and various improvements of it (Hillestad-66) will serve as a means for becoming better acquainted with techniques of modeling and the use of programming languages such as ACSL.

In the simple left heart model shown in lumped circuit form in Fig. 4.2.1a, the atrial pressure, PAT, is assumed to be fixed, as is the central venous pressure, PSV. (Note that the nomenclature here is that used in ACSL programs rather than the subscripted forms used in the diagram. Also, the variable quantities that are often expressed using lowercase letters also are in uppercase.)



Figure 4.2.1. A simple lumped model of the left heart.
(a) Circuit representation of idealized left heart and systemic system load (in cgs units, except pressures in mmHg).
(b) A typical locus (or path) for one cycle of left ventricular pressure, PLV, plotted versus volume, QLV. Note that it lies between the lines whose slopes are the diastolic (minimum) muscle stiffness SLD and the systolic (maximum) stiffness, SLS. End systolic and end-diastolic pressures, PLVES and PLVED, are indicated.
(c) Time variations in stiffness, SLV, plotted versus time for two heart cycles. Initially these pulses will be assumed to be rectangular, and later on more realistic half-sinusoids will be used.

The blood inflow to the left ventricle (FAT) will occur during diastole, when the walls of this chamber are relaxed, and this flow can pass through the mitral valve (MV) and its associated flow resistance, RAT. The ventricle is modeled with a single varying compliance, corresponding to a muscle that is relaxed during the filling part of the heart cycle, or diastole, when stiffness SLV ≈ SLD, and strong during systole, when stiffness is high and SLV ≈ SLS. The variation of SLV with time is approximately as shown in Fig. 4.2.1c, but rectangular pulses will be used initially. The pulsing of SLV results in a counterclockwise locus in ventricular pressure-volume space, as shown by dotted line in Fig. 4.2.1b.

Outflow from the ventricle (FLV) occurs during systole when PLV increases, closing the mitral valve, and then with further increase causes the aortic valve (AV) to open. The resistance of the aortic valve (RLV) will be assumed to be zero because it is small and is in series with the larger resistances of the systemic load that follow.

The systemic arterial impedance or "load" consists of only two resistances and a compliance. The resistance (RSA) is the characteristic impedance of the aorta viewed as a transmission line, and RCA is the total resistance of all body (or systemic) capillary beds in parallel, while CSA is the total compliance of the systemic arterial system. This reduced model of the systemic arterial system is sometimes referred to as the "Westkessel" model (Westerhof-69) and is discussed in more detail in Section 4.4.

The ACSL program LH-PF-1 for simulation study of the left heart model of Fig. 4.2.1a is shown here. CGS units are used except for the applied input pressure, which is given in medical units (PATM = 6 mm Hg); note that this pressure and the systemic venous pressure (PVE) converted to CGS units in the INITIAL part of the program for use in the DERIVATIVE part. In similar fashion, the variable pressures PLV and PCA are converted from CGS to medical units (using PLVM = PV/1332, etc.) in the final part of the DYNAMIC section to make it convenient to plot pressures in the more familiar mm Hg units. It would also be possible to convert to pressures in kilopascals, or to flows in liters/min (see Section 2.6), if so desired.

In this model the parameters are roughly determined as follows. The diastolic filling period is assumed to last for 0.5 sec and the systolic period for 0.3 sec, for a heart period total of 0.8 sec, corresponding to 60/0.8 = 75 beats/min. The fixed pressure PAT at the entrance to the atrium causes FAT to flow through resistance RAT when the mitral valve is open during diastole, filling the ventricle. This resistance is assumed to be 5.0 CGS "fluid ohms," and in combination with the relaxed stiffness of the ventricle has a diastolic time constant of RAT*CLD, or 5/SLD; thus, for SLD = 67, as in the LH-PF-1 program, this time constant is


```
PROGRAM LH-PF-1
  'All quantities in DERIVATIVE are in cgs units'
  'some pressures elsewhere in Medical units, e.g. PATM'
INITIAL
   Constant PATM = 6.0,PVEM = 3.0
   PAT =PATM*1332.
   PVE =PVEM*1332.              $ 'Convert Medical to cgs units'
END $ 'of Initial'
DYNAMIC
  Cinterval CINT= 0.01
  Constant TF = 4.0
 DERIVATIVE
  Algorithm IALG = 4            $ 'Runge Kutta 2'
  Maxterval MAXT =0.005         $ Nsteps NSTP = 1

   'Differential Equations'
  Constant RAT=5.0
  FATX=(PAT-PLV)/RAT            $ 'Atrial flow with no valve.'
  FAT= BOUND(0.0,5000.,FATX)    $ 'Atrial flow with valve.'
  Constant QLVIC=120.,RSA=80.0
  QLV=INTEG((FAT-FLV),QLVIC)    $ 'Left ventricle volume.'
  PLV= QLV*SLV                  $ 'Left ventricle pressure.'
  FLVX=(PLV-PCA)/RSA            $ 'Left ventr. flow, no valve.'
  FLV= BOUND(0.0,1000.,FLVX)    $ 'L. ventr. flow, with valve.'
  Constant QCAIC=220.,CCA=.0022,RCA=1250.
  QCA=INTEG(FLV-FCA,QCAIC)      $ 'Volume, small arteries.'
  PCA= QCA/CCA                  $ 'Capillary entrance press.'
  FCA= (PCA-PVE)/RCA            $ 'Flow into capillaries.'

  Constant SLS=2500.,SLD=67.,TS=0.3,TH=0.8
   'Here TS is length of systole, and TH is the heart period'
  Logical XX
  X = T - ZOH(T,0.,0.,TH)       $ 'Creates sawtooth wave.'
  XX=(X .LE. TS)                $ 'XX is True during systole.'
  SLV=RSW(XX,SLS,SLD)           $ 'Square-wave stiffness out.'
 END $ 'of Deriv.'

  TERMT (T .GE. TF)
  PLVM= PLV/1332.               $ 'Convert cgs output pressures'
  PCAM= PCA/1332.               $  'to medical units.'
 END $ 'of Dynamic'
END $ 'Of Program'
```

0.075 sec, which should allow adequate ventricular filling during diastole.

An average ventricular outflow (cardiac output) of 90 ml/s is assumed; at a heart period of 0.8 sec, this would correspond to a stroke

volume of 90*0.8 = 72 ml. If the ejection fraction of the ventricle is 60 percent, then QLV would have a maximum volume of 72/0.60 = 120 ml and a minimum volume of 120 − 72 = 48 ml. From these volumes and assumptions as to pressures, the ventricular maximum and minimum stiffnesses may be determined. Thus, at the end of diastole

$$\text{SLD} \approx \text{PAT} * 1332/\text{QLV(max)} = 6.0*1332/120 = 66.6 \tag{4.2.1}$$

At the end of systole, if a ventricular pressure of PES = 90 mm Hg is assumed, the systolic (or maximum) stiffness is

$$\text{SLS} \approx \text{PES} * 1332/\text{QLV(min)} = 90.0*1332/48 = 2497.5 \tag{4.2.2}$$

Values close to these calculated normal values of ventricular stiffness, SLS = 2500 and SDS = 67, were chosen for model LH-PF-1, and initial volumes QLVIC = 120 and QCAIC = 220 were chosen after being adjusted to give small starting transients.



Figure 4.2.2. (a) Basic triangular waves, X, which can be used to create various ventricular wave shapes.
(b) Simple rectangular stiffness pulses SLV, obtained from a logic wave XX.

After the constants determined above were introduced into the model program LH-PF-1, various outputs were obtained and plotted using short run times of TF = 3.2 sec, or four beats, after some slight adjustments to give negligible starting transients. The first outputs shown (see Fig. 4.2.2) are for X (a succession of triangular waves of period



Figure 4.2.3. (a) Diastolic inflow, FAT, and ventricular volume, QLV.
(b) Systolic outflow, FLV, load flow, FCA, and QLV.
(c) Ventricular pressure, PLV, and arteriolar pressure, PCA, shown in medical units as PLVM and PCAM.
(d) Locus on the PLVM versus QLV plane.

0.8 sec.) and the square-wave ventricular stiffness SLV, obtained using the logic signal XX = (X .LE. TS).

The diastolic inflow, FAT, and the ventricular volume, QLV, are shown in Fig. 4.2.3a and the outflow, FLV, and QLV in Fig. 4.2.3b. The sharp rise to a peak and exponential drop in each flow pulse are not natural and will be replaced by more realistic waveforms when the model is improved with more rounded SLV pulses, and when inertances are included. Such an improved model is presented later in this section.

Figure 4.2.3c shows plots of the ventricular pressure, PLV, and the load pressure, PCA (located approximately in the arterioles). Note the slow exponential decay of PCA during diastole; the time constant here is TCA = RCA*CCA = 2.75 sec, and thus during the 0.5 sec of diastole, the pressure may be approximated by

$$\begin{aligned} PCA &= PES*Exp(-T/TCA) \\ &\approx PES*(1.0 - T/2.75) \end{aligned} \tag{4.2.3}$$

where PES is the pressure at the end of systole. An examination of Fig. 4.2.3c shows that PES ≈ 84.5 and droops to about 71 at the end of diastole, which agrees fairly well with Eq. (4.2.3).

Another very important kind of plot is the locus of ventricular pressure (PLV) plotted versus volume (QLV), as shown in Fig. 4.2.3d; here the straight-line curves of slopes SLS and SLD may be seen as parts of the locus, and the counterclockwise direction of the locus has been indicated. Note that the width of this locus is the stroke volume. More realism would require rounded SLV pulses (as in Fig. 4.2.1c) and an upward curving diastolic stiffness curve for increasing QLV.

Because the system is piecewise linear, it is possible to use Laplace transform methods to obtain a numerical solution. This would require finding initial conditions and starting a new solution each time a valve opened or closed. Problems would mount up if detail were added to improve the model, and the addition of even one important nonlinearity would make Laplace solutions impossible. Thus a time-domain solution using a program such as ACSL is usually regarded as the best procedure for simulation study of system models, even at low levels of complexity.

Improvements will now be made in model LH-PF-1, and these changes will be incorporated in models of the complete cardiovascular loop. The first improvement is to change the actuating signal from a square-topped wave to a clipped half-sinusoid and to design this actuator to be one that can be used with the feedback systems of the body to change factors such as heart rate. This is achieved in the ACSL program shown below by again subtracting a zero-order hold (ZOH) from time T each TH seconds to generate a series of sawtooth waves (X) of unit slope, as in the simple generator of LH-PF-1. Then a logic signal XX is set to

TRUE whenever the repeating sawtooth X is of amplitude less than systolic time TS. The logic signal XX is used in an ACSL Real-Switch (RSW) to generate a spaced triangular wave (STW) during the systolic period. This wave, in turn, is used to generate spaced sine waves, (SSW).



Figure 4.2.4. (a), (b) Waveforms in an improved activity generator.
(c) Ventricular and systemic pressures, PLVM and PCAM, and volume, QLV, with a half-sine-wave activity generator.
(d) Ventricular pressure-volume locus with an improved activity generator.

```
LOGICAL XX
Constant PI = 3.14159
X = T - ZOH(T,0.,0.,TH)
XX=(X .LE. TS)
STW=RSW(XX,X,0.0)
SSW=SIN(PI*STW/TS)
ACT=BOUND(0.,1.0,1.15*SSW)
SLV =SLD*(1.-ACT) + SLS*ACT
```

The spaced sine waves are amplified by 1.15, then clipped back to unity to give the activity function, ACT, which is used in turn to give a stiffness, SL, with near-half-sinusoid systolic pulse forms. These various waveforms may be obtained using the set of eight commands listed above in place of the last four commands in the DERIVATIVE section of LH-PF-1. This gives a new program that we will call LH-PF-2, which yields the waveforms shown in Figs. 4.2.4a and 4.2.4b.

This new actuator or activity generator may be modulated in period, systolic duration, and amplitude, and thus can be quite useful in a model that includes feedback control to maintain blood pressure levels when blood volume, flow resistances, or compliances change. Here these parameters are fixed, but it can be seen in Fig. 4.2.4c that ventricular pressure and flow waveforms have been improved. More impressive is the improvement in the shape of the ventricular pressure versus volume locus, as shown in Fig. 4.2.4d.

Two other improvements will now be made in the model, as shown in lumped circuit form in Fig. 4.2.5. These changes involve the introduction of inertance terms. One of these, L_{AO}, is included in a PI-section added between the aortic valve and the systemic load used in LH-PF-1; it also includes a resistance RAO and compliances CAO and CSA. This added section provides a better model of the first part of the aorta after the aortic valve. The equation for the flow through RAO and LAO is of the form given in (4.1.15). In ACSL notation this becomes

```
FAO = INTEG((PAO-PSA - RAO*FAO)/LAO, FAOIC)
```

(4.2.3)

The inertance LLV must be dealt with in a different way because it is in series with the aortic valve. It is not correct to use an integrator as in (4.2.3), followed by a separate limiter or BOUND command, because this might cut off the flow FAO when it should continue to appear due to inertial effects. To prevent this effect, often called "integrator windup," a combination integrator and limiter must be used, called the LIMINT command in ACSL. Thus FLV is now obtained by using


```
FLV= LIMINT((PAO-PSA -RAO*FAO)/LAO,FLVIC,0.0,5000.
```

(4.2.4)

Another small improvement in the spaced sine-wave SSW results because a small second-harmonic term is used to provide some "skew" in this waveform; this is also used in a following model, PF-1.

The complete ACSL program for the model LH-PF-3 shown in Fig. 4.2.5 is given next. Note that most constants are now included in the DERIVATIVE section, located near the point where they first appear for convenience.

```
PROGRAM LH-PF-3
    Constant PATM=6.0, PVEM=3.0
INITIAL
    PAT= PATM*1332.       $ PVE= PVEM*1332.
END $ 'of Initial'
DYNAMIC
   Cinterval CINT= 0.01
   Constant TF=4.0
 DERIVATIVE
   Algorithm IALG = 5     $ 'Runge Kutta 4'
```



Figure 4.2.5. (a) Lumped-circuit equivalent for model LH-PF-3. (b) Ventricular stiffness, SLV, plotted versus time.

```
    Maxterval MAXT =0.001 $ Nsteps NSTP = 1
    LOGICAL XX
   Constant TH=0.8,TS=0.3,SLD=67.,SLS=2400.0
   Constant PI=3.14159,K1=.9,K2=0.3,B=1.05
    X = T - ZOH(T,0.,0.,TH)
    XX=(X .LE. TS)
    STW= RSW(XX,X,0.0)
    SSW= K1*SIN(PI*STW/TS)-K2*SIN(2.*PI*STW/TS)
    ACT= BOUND(0.,1.0,B*SSW)
    SLV= SLD*(1.-ACT) + SLS*ACT
   Constant RAT=5., QLVIC= 120., QLVU=0.0
    FAT= BOUND(0.0,5000.,(PAT- PLV)/RAT)
    QLV= INTEG((FAT-FLV),QLVIC)
    PLV= (QLV-QLVU)*SLV
   Constant RLV=5.0,LLV=0.5,RA0=5.,LA0=.5, . . .
    CA0=.00015, QA0IC=100., QA0U=85.
    FLV= LIMINT((PLV-PA0)/LLV - RLV *FLV/LLV,0.,0.,5000.)
    QA0= INTEG((FLV-FA0),QA0IC)
    PA0= (QA0=QA0U)/CA0
    FA0= INTEG((PA0-PSA)/LA0 - RA0*FA0/LA0,0.0)
   Constant CSA= .0003, RSA= 50.,QSAIC=281.,QSAU=250.
    QSA= INTEG(FA0-FSA,QSAIC)
    PSA= (QSA-QSAU)/CSA
    FSA= (PSA-PCA)/RSA
   Constant CCA=0.0022, QCAIC=1010.,QCAU=810.,RCA=1150.
    QCA= INTEG(FSA-FCA,QCAIC)
    PCA= (QCA-QCAU)/CCA
    FCA= PCA/RCA
  END $ 'of Deriv.'
    PLVM= PLV/1332.
    PAOM= PA0/1332.
    PSAM= PSA/1332.
    PCAM= PCA/1332.
    TERMT (T .GE. TF)
    Q= QLV + QA0 + QSA + QCA
  END $ 'of Dynamic'
END  $ 'of Program'
```

Some runs made with LH-PF-3 are shown in Fig. 4.2.6. It can be seen that the pressure within the ventricle is still somewhat peaked, but that some oscillations appear near the end of systole. The aortic pressure PAO follows PLV during systole, and PCA is much more rounded. During diastole, PAO and PCA nearly coincide. Note, however, that the oscillations in these waves are of a higher frequency than is normally observed, and that no dicrotic notch at the end of systole is observed. The flow pulses, also shown in Fig. 4.2.6b, are improved but are still not very



Figure 4.2.6. Pressure and flow waveforms obtained with the left ventricular model LH-PF-3. (a) Left ventricular flow FLV and aortic flow FAO. The latter shows the oscillations which occur in the aorta just after valve closure.
(b) Left ventricular pressure, PLVM, and superimposed aortic pressure PAOM, both in mmHg.
(c) Aortic flow FSA and the much less pulsatile capillary flow FCA.
(d) Left ventricle pressure-versus-volume locus. The height of this loop gives the peak value of PLVM, and the width gives the stroke volume.

realistic. It will be demonstrated later that even a slightly more detailed multisection systemic model will give more satisfactory realism in waveforms.

The slow descent of flow FCA during diastole corresponds approximately to the first part of a decaying exponential of time constant TCA $\approx$ RCA*CCA = 2.75 sec (see Fig. 4.2.5a). As an approximate check, note that the peaks of FCA are at about 125 ml/s, and diastolic delay lasts about 0.6 sec. The lowest points in FCA should be given by

$$\begin{aligned} \mathrm{FCA}_{\min} &= 125.0*\exp(-T/\mathrm{TCA}) \\ &\approx 125.*(1. - 0.6/2.75) = 98.\ \mathrm{ml/s} \end{aligned}$$

which is fairly close to the value observed. Simple checks of this kind are important to verify that the model is obeying the equations used.